DESCRIPTION OF PROGRAM

PURPOSE

There are many instances in which experimental data are available but no theoretical equation has been derived for expressing the relationship between two physical parameters. Sometimes the theoretical equation is known but is too complicated or cumbersome for use in mathematical evaluation. A polynomial is often used to approximate this relationship. If the coefficients of the polynomial are determined from the available data by the least squares procedure, the resulting equation can be used to evaluate the dependent variable with statistical validity from given values of the independent variable.

This program generates an approximating polynomial by the least squares technique. The equation so derived contains as many terms as necessary to bring the standard error of the dependent variable within a range specified by the user.

GENERAL DESCRIPTION

Provision has been made for a maximum of 100 observations and a polynomial expansion up to the 15th degree.

This program is based on the assumption that a set of experimental data can be fitted to a polynomial of the form:

$Y = AO + A1x + A2x^2 \ldots\ldots\ldots\ldots\ldots + A15x^{15}$.

The linear form $Y = AO + A1x$ is tried as a first approximation. The coefficients (A's) are computed and the standard error of the dependent variable (Y) is compared to a predetermined maximum value. If the error is greater than the maximum value, the process is repeated, adding a term of the form $AN\ x^N$ (where N = 2, 3, ... , 15) until the error is within the tolerance. In each case the coefficients are computed by the least squares technique. The calculations utilize FORTRAN floating point arithmetic. A modified Gaussian elimination technique is used to solve the resulting set of linear equations.

INPUT FORMAT

All input to the program must conform to the specifications for fixed and floating point constants as outlined in the 1620 FORTRAN Manual. The input to the program consists of two types of records. The first record is read into the 1620 by Fortran Statement 1 which is "READ, N, TOL, LAST." The next N records are the observations and are read into the 1620 by statement 20 "READ, X(I), Y(I)" or if SW1 is on by statement 30 "READ, X(I), Y(I), W(I)." See page 1 of the appendix for explanation of the symbols used. The

exact format of these N records is dependent upon the setting of SW1. It is important to note that data are always entered with one observation to the record.

If after obtaining a complete output, it is desired to enter a new TOL and/or LAST, this can be accomplished through the console typewriter under control of SW4. See Operating Notes for details.

OUTPUT

All output is presently on the console typewriter. This can be easily modified by changing any of the PRINT statements to PUNCH statements. SW2 and SW3 control the format of the output.

SW2 ON causes the following format to be printed for each order polynomial fitted:

N O R D	T O L	S 2	N
0	COEFFICIENT		
1	COEFFICIENT		
⋮	⋮		
NORD	COEFFICIENT		

SW3 ON causes the following additional output to be printed provided SW2 is also ON:

X(1)	Y(1)	$\hat{Y}(1)$	$Y(1) - \hat{Y}(1)$
⋮	⋮	⋮	⋮
X(N)	Y(N)	$\hat{Y}(N)$	$Y(N) - \hat{Y}(N)$

All output will be obtained for the order polynomial being fitted if NORD = LAST or if S2 ≤ TOL. If the output occurs under either or both of these conditions, the machine will halt after printing. See Operating Notes for switch settings before depressing START.

MACHINE REQUIREMENTS

Any IBM 1620 for which a FORTRAN compiler has been written.

TABLE OF CONTENTS

Author:

W. R. Graves
IBM Corporation
Mohansic System Laboratory
Advanced Systems Development Division
2651 Strang Boulevard
Yorktown Heights, New York 10588

FILE NO. ABSTRACT

07.0.002 IBM 1620-POLYNOMIAL CURVE FITTING *CARD*
AVAILABLE 1ST QUARTER 1962

AUTHOR- **W. R. Graves**
IBM Corporation
Mohansic System Laboratory
Advanced Systems Development Division
2651 Strang Boulevard
Yorktown Heights, New York 10588
DIRECT INQUIRIES TO AUTHOR

THIS PROGRAM GENERATES AN APPROXIMATING POLYNOMIAL BY THE LEAST SQUARES TECHNIQUE. THE EQUATION SO DERIVED CONTAINS AS MANY TERMS AS NECESSARY TO BRING THE STANDARD ERROR OF THE DEPENDENT VARIABLE WITHIN A RANGE SPECIFIED BY THE USER, OR TO FIT A 15TH ORDER POLYNOMIAL. PRINTING OF INTERMEDIATE COEFFICIENTS AND THE PRINTING OF A TABULATION OF OBSERVED VS CALCULATED VALUES OF THE DEPENDENT VARIABLE ARE UNDER THE CONTROL OF PROGRAM SWITCHES AS IS THE INCLUSION OF WEIGHTING FACTORS. THE CALCULATIONS UTILIZE FLOATING ARITHMETIC WITH AN 8 DIGIT MANTISSA. IBM 1620, 20K CORE, 1622 CARD READ-PUNCH.

OPERATING NOTES

1. Continue to Next Order Polynomial when Computer Halts after Printing.

   Turn SW4 ON
   Depress START
   When typewriter is activated, key in a value for TOL and LAST followed by a ≠.
   Turn SW4 OFF
   Depress RELEASE
   Depress START

   Note: If error is made while keying in this data, depress RELEASE and Start with SW4 ON. This will cause the typewriter to be activated and the whole record must be keyed in again.

2. ERROR E7 OCCURS.

   You have probably divided by zero in the calculation of S2 between statement 160 and statement 161. The reason for this is that the order of the polynomial is within 1 of the number of observations. The coefficients calculated here will be correct, but the standard error S2 will be distorted. If you continue to the next order polynomial, ERROR F6 will occur as a result of taking the square root of a negative number when calculating S2.

3. If it is desired to compare $\sum_{i=1}^{N} (Yi - Yi)^2$ rather than the present S2 against TOL, delete the two FORTRAN statements immediately following statement 160 and re-assemble the program.

4. Most questions about the program can be resolved by reference to the accompanying flow chart and FORTRAN listing. A complete analysis of the actual storage requirements of the program has not been made. The program did exceed memory by 1011 positions when assembled with the trace routine so there is very little space available for additions or lengthy modifications.

5. A considerable amount of storage could be gained by deleting that portion of the program dealing with weight factors.

OPERATING INSTRUCTIONS CARD SYSTEM

I. Clear Memory
   RESET, INSERT, 310000300002, RELEASE, START
   INSTANT STOP

II. Ready Typewriter for Normal Fortran Output
   Margins 10 and 95.
   Tabs every 17 positions.

III. Load Program
   Ready 1622 Read Hopper with Program Deck
   RESET - 1620
   LOAD - 1622
   "LOAD DATA" will type out after loading is completed.

IV. Ready Paper in Typewriter

V. Place Data Cards in 1622 Read Hopper

VI. Set Console Switches
   OVERFLOW CHECK SWITCH - PROGRAM
   ALL OTHER CHECK SWITCHES - STOP
   PROGRAM SWITCHES:

	ON	OFF
1	Weight Factors	No Weight Factors
2	Print Polynomial Coefficients For Each Order Fitted	List Coefficients Only When NORD = LAST or S2 ≤ TOL.
3.	Print Table of Observed vs Calculated After Listing Coefficients.	Print Table of Observed vs Calculated Only When NORD = LAST or S2 ≤ TOL.
* 4	Continue Present Problem (Normally off)	Read Next Set of Data

   * See pp. 3 for additional information on SW4.

VII. Depress START ---- Or INSERT 4907500.

APPENDIX

SYMBOLS USED IN FORTRAN PROGRAM

X(I)	i^{th} value of independent variable.
Y(I)	i^{th} value of dependent variable.
W(I)	Weight factor of i^{th} observation.
A(I, J)	Matrix of coefficients of normal equations.
N	Number of observations.
TOL	Allowable value of standard error.
LAST	Highest order polynomial to be fitted.
Sum X(J)	$\sum_{i=1}^{N} (Wi)\,(Xi)^{J-1}$
Sum Y(J)	$\sum_{i=1}^{N} (Yi)\,(Yi)\,(Xi)^{J-1}$
NORD	Order polynomial fitted.
S1	Calculated value of Yi using Xi (also called $\hat{Y}i$).
S2	$\sqrt{\dfrac{\sum_{i=1}^{N} (Yi - \hat{Y}i)^2}{(N - NORD - 1)}}$
S3	$Yi - \hat{Y}i$

POLYNOMIAL CURVE FITTING
FLOW CHART





```
C  POLYNOMIAL CURVE FITTING
C
C  W R GRAVES  IBM NEW ORLEANS
C
      DIMENSION X(100),Y(100),A(16,16),SUMX(31),SUMY(15),W(100)
C
    1 READ,N,TOL,LAST
      DO 40 I=1,N
      IF(SENSE SWITCH 1)30,20
   20 READ,X(I),Y(I)
      GO TO 40
   30 READ,X(I),Y(I),W(I)
   40 CONTINUE
      IF(SENSE SWITCH 1)70,50
   50 DO 60 I=1,N
   60 W(I)=1.
   70 SUMX(1)=0.
      SUMX(2)=0.
      SUMX(3)=0.
      SUMY(1)=0.
      SUMY(2)=0.
      DO 90 I=1,N
      SUMX(1)=SUMX(1)+W(I)
      SUMX(2)=SUMX(2)+W(I)*X(I)
      SUMX(3)=SUMX(3)+W(I)*X(I)*X(I)
      SUMY(1)=SUMY(1)+W(I)*Y(I)
   90 SUMY(2)=SUMY(2)+W(I)*X(I)*Y(I)
      NORD=1
   91 IF(SENSE SWITCH 4)92,93
   92 ACCEPT,TOL,LAST
   93 L=NORD+1
      KK=L+1
      DO 101 I=1,L
      DO 100 J=1,L
      IK=J-1+I
  100 A(I,J)=SUMX(IK)
  101 A(I,KK)=SUMY(I)
      DO 140 I=1,L
      A(KK,I)=-1.
      KKK=I+1
      DO 110 J=KKK,KK
  110 A(KK,J)=0.
      C=1./A(1,I)
      DO 120 II=2,KK
      DO 120 J=KKK,KK
  120 A(II,J)=A(II,J)-A(1,J)*A(II,I)*C
      DO 140 II=1,L
      DO 140 J=KKK,KK
  140 A(II,J)=A(II+1,J)
      S2=0.
      DO 160 J=1,N
      S1=0.
      S1=S1+A(1,KK)
      DO 150 I=1,NORD
  150 S1=S1+A(I+1,KK)*X(J)**I
  160 S2=S2+(S1-Y(J))*(S1-Y(J))
      B=N-L
      S2=(S2/B)**.5
      IF(SENSE SWITCH 2)163,161
  161 IF(NORD-LAST)162,163,162
  162 IF(S2-TOL)163,163,171
  163 PRINT,
      PRINT, NORD, TOL, S2, N
      DO 164 I=1,L
      J=I-1
  164 PRINT,J,A(I,KK)
      IF(SENSE SWITCH 3)167,165
  165 IF(NORD-LAST)166,167,166
  166 IF(S2-TOL)167,167,171
  167 DO 169 I=1,N
      S1=0.
      S1=A(1,KK)
      DO 168 J=1,NORD
  168 S1=S1+A(J+1,KK)*X(I)**J
      S3=Y(I)-S1
  169 PRINT,X(I),Y(I),S1,S3
      IF(NORD-LAST)170,173,173
  170 IF(S2-TOL)173,173,171
  171 NORD=NORD+1
      J=2*NORD
      SUMX(J)=0.
      SUMX(J+1)=0.
      SUMY(NORD+1)=0.
      DO 172 I=1,N
      SUMX(J)=SUMX(J)+X(I)**(J-1)*W(I)
      SUMX(J+1)=SUMX(J+1)+X(I)**J*W(I)
  172 SUMY(NORD+1)=SUMY(NORD+1)+Y(I)*X(I)**NORD*W(I)
      GO TO 91
  173 PAUSE
      IF(SENSE SWITCH 4)171,1
      END
```

SAMPLE PROBLEM 1 NO WEIGHT FACTORS

1	1.0000000E-03	4.7140470E-08	11
0	2.0000000		
1	-.99999999		
-5.0000000	7.0000000	7.0000000	.00000000
-4.0000000	6.0000000	6.0000000	.00000000
-3.0000000	5.0000000	5.0000000	.00000000
-2.0000000	4.0000000	4.0000000	.00000000
-1.0000000	3.0000000	2.9999999	1.0000000E-07
.00000000	2.0000000	2.0000000	.00000000
1.0000000	1.0000000	1.0000001	-1.0000000E-07
2.0000000	.00000000	.00000000	.00000000
3.0000000	-1.0000000	-1.0000000	.00000000
4.0000000	-2.0000000	-2.0000000	.00000000
5.0000000	-3.0000000	-3.0000000	.00000000

11

SAMPLE PROBLEM 2 NO WEIGHT FACTORS

1	.00000000	.13709614	35
0	.67653964		
1	4.5094777E-02		
2	.00000000	6.8247925E-02	35
0	.88567484		
1	-5.8891750E-02		
2	7.3669823E-03		
3	.00000000	2.2010479E-02	35
0	1.0101370		
1	-.16878278		
2	2.7903453E-02		
3	-9.6864043E-04		
4	.00000000	1.2667654E-02	35
0	1.0527606		
1	-.22962000		
2	4.8020588E-02		
3	-3.1670640E-03		
4	7.5153401E-05		
5	.00000000	1.2528518E-02	35
0	1.0456017		
1	-.21550438		
2	4.1142057E-02		
3	-1.8939045E-03		
4	-2.2981124E-05		
5	2.6586468E-06		
6	.00000000	1.4404736E-02	35
0	1.0549349		
1	-.23916686		
2	5.6848405E-02		
3	-6.1416738E-03		
4	5.1292009E-04		
5	-2.8836416E-05		
6	6.9676638E-07		
7	.00000000	1.1866621E-02	35
0	1.0419582		
1	-.20453217		
2	3.3409581E-02		
3	3.1708200E-05		
4	-1.8186513E-04		
5	-1.2040700E-06		
6	1.0740927E-06		
7	-3.6192355E-08		
8	.00000000	1.2140386E-02	35
0	1.0428216		
1	-.20598632		
2	3.3852269E-02		
3	6.4796813E-05		
4	-2.0604440E-04		
5	7.8117300E-07		
6	1.1782136E-06		
7	-5.4157920E-08		
8	5.5021779E-10		
9	.00000000	1.3673123E-02	35
0	1.0537794		
1	-.22739792		
2	4.3658105E-02		
3	-1.0574389E-03		
4	-3.6231047E-04		
5	3.7879753E-05		
6	4.8058560E-07		
7	-3.4203367E-07		
8	2.3551077E-08		
9	-5.2506785E-10		

12

PARTIAL OUTPUT

SAMPLE PROBLEM 3 NO WEIGHT FACTORS

1	.30000000	2.3394379	11
0	3.0090909		
1	1.8918182		
2	.30000000	1.3773300	11
0	1.0160840		
1	1.8918182		
2	.19930069		
3	.30000000	3.1790805E-02	11
0	1.0160840		
1	1.0097709		
2	.19930069		
3	4.9553224E-02		
-5.0000000	-5.2500000	-5.2444062	-5.5938000E-03
-4.0000000	-3.0000000	-3.0055949	5.5949000E-03
-3.0000000	-1.5500000	-1.5574595	7.4595000E-03
-2.0000000	-.60000000	-.60268089	2.6808900E-03
-1.0000000	.15000000	.15606057	-6.0605700E-03
.00000000	1.0000000	1.0160840	-1.6084000E-02
1.0000000	2.2500000	2.2747087	-2.4708700E-02
2.0000000	4.3000000	4.2292542	7.0745800E-02
3.0000000	7.1500000	7.1770399	-2.7039900E-02
4.0000000	11.400000	11.415384	-1.5384000E-02
5.0000000	17.250000	17.241608	8.3920000E-03